﻿/* Syntax to reproduce analyses reported in Study 2a of "The Inherence Heuristic as a Source of Essentialist Thought"


/* Numbers of cases excluded for failing catch items and non-US IP addresses

FREQUENCIES VARIABLES=Exclude
  /ORDER=ANALYSIS.


/* Exclude participants who (1) failed more than 1 catch item or (2) had a non-US IP address

USE ALL.
COMPUTE filter_$=(Exclude = 0).
VARIABLE LABELS filter_$ 'Exclude = 0 (FILTER)'.
VALUE LABELS filter_$ 0 'Not Selected' 1 'Selected'.
FORMATS filter_$ (f1.0).
FILTER BY filter_$.
EXECUTE.


/* Get decriptives for scale scores and save standardized values

DESCRIPTIVES VARIABLES=IHAvg IHCatchAvg IHCWrongTotal RPMTestSum NoAvg SDOAvg ConfomAvg
  /SAVE
  /STATISTICS=MEAN STDDEV MIN MAX.


/* Table 7 -- Cronbach's alpha for No et al Lay Theory of Race Scale

RELIABILITY
  /VARIABLES=No01Abili No02Cultu No03Ingra No04Basic No05r No06r No07r No08r
  /SCALE('Lay Theory of Race') ALL
  /MODEL=ALPHA.


/* Table 7 -- Cronbach's alpha for Inherence Heuristic Scale

RELIABILITY
  /VARIABLES=IH01Stop IH02Beds IH03Diamonds IH04Dollar IH05Chipmunk IH06OJ IH07Pink IH08White 
    IH09Black IH10Mint IH11Seven IH12Weekend IH13Limbs IH14Eyes IH15Sounds
  /SCALE('Inherence Heuristic Scale') ALL
  /MODEL=ALPHA.


/* Table 7 -- Cronbach's alpha for Raven's Progressive Matrices

RELIABILITY
  /VARIABLES=RPM01score RPM02score RPM03score RPM04score RPM05score RPM06score RPM07score 
    RPM08score RPM09score RPM10score RPM11score RPM12score
  /SCALE("Raven's Progressive Matrices") ALL
  /MODEL=ALPHA.


/* Table 7 -- Cronbach's alpha for Social Dominance Orientation

RELIABILITY
  /VARIABLES=SDO01NotEq SDO02Worth SDO03CareL SDO04Deser SDO05MoreC SDO06Infer SDO07StepO SDO08r 
    SDO09r SDO10r SDO11r SDO12r SDO13r SDO14r
  /SCALE('Social Dominance Orientation') ALL
  /MODEL=ALPHA.


/* Table 7 -- Cronbach's alpha for Conformity Scale

RELIABILITY
  /VARIABLES=Con01 Con02r Con03 Con04 Con05 Con06 Con07r Con08 Con09r Con10 Con11r
  /SCALE('Conformity Scale') ALL
  /MODEL=ALPHA.


/* Table 7 -- Correlations among scale scores
/* Confidence intervals for correlations were calculated with Cumming's ESCI, available at http://www.latrobe.edu.au/psy/research/cognitive-and-developmental-psychology/esci

CORRELATIONS
  /VARIABLES=NoAvg IHAvg RPMTestSum SDOAvg ConfomAvg
  /PRINT=TWOTAIL NOSIG
  /MISSING=PAIRWISE.


/* Regress Lay Theory of Race scores onto Inherence Heuristic Scale, Raven's Progressive Matrices, 
/* Social Dominance Orientation, and Conformity Scale (all standardized) -- Table 6
/* Plus additional model including Catch Item average as additional predictor

REGRESSION
  /MISSING LISTWISE
  /STATISTICS COEFF OUTS CI(95) R ANOVA
  /CRITERIA=PIN(.05) POUT(.10)
  /NOORIGIN 
  /DEPENDENT ZNoAvg
  /METHOD=ENTER ZIHAvg ZRPMTestSum ZSDOAvg ZConfomAvg
  /METHOD=ENTER ZIHCatchAvg.


/* Partial correlations among measured variables reported in Table 7

PARTIAL CORR
  /VARIABLES=NoAvg IHAvg BY RPMTestSum SDOAvg ConfomAvg
  /SIGNIFICANCE=TWOTAIL
  /MISSING=LISTWISE.

PARTIAL CORR
  /VARIABLES=NoAvg RPMTestSum BY IHAvg SDOAvg ConfomAvg
  /SIGNIFICANCE=TWOTAIL
  /MISSING=LISTWISE.

PARTIAL CORR
  /VARIABLES=NoAvg SDOAvg BY IHAvg RPMTestSum ConfomAvg
  /SIGNIFICANCE=TWOTAIL
  /MISSING=LISTWISE.

PARTIAL CORR
  /VARIABLES=NoAvg ConfomAvg BY IHAvg RPMTestSum SDOAvg 
  /SIGNIFICANCE=TWOTAIL
  /MISSING=LISTWISE.
